# Computable General Equilibrium Models: Income generation, distribution and taxation

Wolfgang Britz, Institute for Agricultural Policy, University Bonn

Simulation Models for Policy Analysis

Summer Term 2020

# Aims for today

- Learn how income generation, distribution and taxation are typically implemented in CGEs
- Understand pro and cons of the regional household approach
- Learn about welfare analysis under the regional household approach

# Income generation

- Income in CGEs stems from three sources:
  - Factor income
  - Income from taxing economic transactions (output, factor use, intermediate and final consumption, imports and exports, factor income)
  - Income transfers such as foreign savings

# Income distribution

- Two competing approaches:
  1. Regional household approach (GTAP)
  2. Separate account for institutions

# Income distribution

- Regional household approach (GTAP)
  - One central collection point for factor income and all taxes
  - From there distribution to the final demand of the different institutions (private, government, investment) via CD or similar ⇔ social welfare function!
  - Simple, requires only limited data
  - Can provoke counter-intuitive results: lowering taxes does not mean that the government will spent less, as government receives a share of total income (there is no separate government budget which links government income and government spent)

# Income distribution

- Separate account for institutions
  - Each institution has its own closed account (income, savings, expenditures)
  - Requires more data and assumptions (e.g. will less tax income for the government mean less consumption or less government savings?)

# Taxation

- In our toy model only
  1. Factor(=income tax), tax rates differ by factor

```
v_yTax(R,"f")  =E= SUM( (f,a),    v_xf(r,f,a) * v_pfa(r,f,a) * p_fTax(r,f));
```

  2. Output use taxes

```
e_ytaxS(R,"s") ..

  v_yTax(R,"s")  =E= SUM(s_to_c(s,c), v_x(r,s) * v_px(r,c) * p_oTax(r,c));
```

# Regional household



# Welfare analysis

1. Utility functions define aggregate quantity indices

```
p_res(r,"gov","u","v",%1)
    = prod(c $ p_alphaa(r,c,"gov"),
          v_xa.l(r,c,"gov")**p_alphaa(r,c,"gov") );

p_res(r,"inv","u","v",%1)
    = prod(c $ p_alphaa(r,c,"inv"),
          v_xa.l(r,c,"inv")**p_alphaa(r,c,"inv") );

p_res(r,"hou","u","v",%1)
    = prod(c $ p_alphaa(r,c,"hou"),
          (v_xa.l(r,c,"hou")-v_gamma.l(r,c))**p_alphaa(r,c,"hou") );
```

2. Define price indices (expenditure / quantity index)

```
p_res(r,"gov","p","v",%1) = v_yg.l(r) / p_res(r,"gov","u","v",%1);
p_res(r,"inv","p","v",%1) = v_ys.l(r) / p_res(r,"inv","u","v",%1);
p_res(r,"hou","p","v",%1) = v_yc.l(r) / p_res(r,"hou","u","v",%1);
```

# Welfare analysis

3. Calculate equivalent variation: minimum expenditure to reach utility in simulation under the prices of the benchmark, minus benchmark expenditure

```
p_res(r,"gov","ev","v",%1) =
    + p_res(r,"gov","u","v",%1)
      * prod(c $ p_alphaa(r,c,"gov"),
             (1/p_alphaa(r,c,"gov"))**p_alphaa(r,c,"gov") )
  - p_res(r,"gov","e","v","bench");

p_res(r,"inv","ev","v",%1) =
    + p_res(r,"inv","u","v",%1)
      * prod(c $ p_alphaa(r,c,"inv"),
             (1/p_alphaa(r,c,"inv"))**p_alphaa(r,c,"inv") )
  - p_res(r,"inv","e","v","bench");

p_res(r,"hou","ev","v",%1) =
      sum(c, v_gamma.l(r,c))
    + p_res(r,"hou","u","v",%1)
      * prod(c $ p_alphaa(r,c,"hou"),
             (1/p_alphaa(r,c,"hou"))**p_alphaa(r,c,"hou") )
  - p_res(r,"hou","e","v","bench");
```

# Welfare analysis

4. Aggregate over institutions:

```
p_res(r,"regy","ev","v",%1) = sum(dem, p_res(r,dem,"ev","v",%1));
```

EV summarizes for each institution and for the economy as a whole the welfare change (under the regional household approach), measured in money terms

# myGTAP

- myGTAP was developed to overcome the limitation of the regional household approach in GTAP and allow for modeling several private household types
- Considers also remittances (= income exchanges with foreign household), foreign factor income and income transfers with other household in same country
- Requires additional information not comprised in standard GTAP SAM:
  - Share on factor income for each household type
  - Share on direct taxes of each household type
  - How income is spent (consumption of different product, savings) by each household types

# myGTAP: Factor income and direct taxes

```
*
* --- Factor income by households, depends on ownership shares shrXF on factor supply xft / xf
*
 factYHeq(rs(r),h,ts(t)) $ hFlag(r,h) ..

   factYH(r,h,t)/regy.scale(r,t) =e= [   sum( (f,a) $ xfFlag(r,f,a), pf(r,f,a,t)*xf(r,f,a,t)*shrXf(r,h,f,a,t))
                                       - fdepr(r,t)*pi(r,t)*kstock(r,t)*shrDep(r,h,t) ] / regy.scale(r,t);
*
* --- Direct taxation of household, depends on ownership shares shrXF on factor supply xft / xf
*     and household specific direct tax rates kappaFH
*
 dirTaxEq(rs(r),f,h,ts(t)) $ hFlag(r,h) ..

   dirTax(r,f,h,t)/regy.scale(r,t) =E=
            sum( a $ xfFlag(r,f,a),  kappafH(r,f,h,t) * pf(r,f,a,t)*xf(r,f,a,t)*shrXf(r,h,f,a,t))/regy.scale(r,t);
```

Income tax rate  Factor income  share

Note:

- mnemonics slightly different compared to toy model
- GTAP model comprises depreciation (rate: fdepr, pi: price of investments, kstock: physical capital stock, shrDep: share allocated to household)

# myGTAP: total household income

```
*
* --- household income definition
*
 hInceq(rs(r),h,ts(t)) $  sum(f,shrXf(r,h,f,t)) ..

   hInc(r,h,t)/regy.scale(r,t) =E=[  factYH(r,h,t)
                  + trng(r,h,t)
                  + remiH(r,h,t)*remihShrCor(t)*lcu(r,t)
                  + fyiH(r,h,t)*fyihShrCor(t)*lcu(r,t)
                  + sum(h0, trnh(r,h0,h,t))
                  - sum(f, dirTax(r,f,h,t))
                  - remoH(r,h,t)*lcu(r,t)
                  - fyoH(r,h,t)*lcu(r,t)
                  - sum(h0, trnh(r,h,h0,t))]/regy.scale(r,t);
```

Transfer from abroad

Transfer from government

Foreign factor income

From other households

Direct taxes

To foreign household

To foreign factor income

To other households